# $Id$
# Maintainer: Felix Yan <felixonmars@archlinux.org>

pkgname=perl-mailtools
pkgver=2.19
pkgrel=2
pkgdesc="Various e-mail related modules"
arch=('any')
url="https://metacpan.org/release/MailTools"
license=('PerlArtistic' 'GPL')
depends=('perl-timedate')
options=('!emptydirs')
source=("http://search.cpan.org/CPAN/authors/id/M/MA/MARKOV/MailTools-$pkgver.tar.gz")
sha512sums=('a4ecba2acba56f2d41294e1b299c08b138d5e4796880635afcea7628adf60e11d5b138fc56001c331f765048bd91fa94da27bd7fdeedc51ff91e4ef5770a068e')

build() {
  cd MailTools-$pkgver
  perl Makefile.PL INSTALLDIRS=vendor
  make
}

check() {
  cd MailTools-$pkgver
  make test
}

package() {
  cd MailTools-$pkgver
  make DESTDIR="$pkgdir" install
}
